
1. Overview

This document lists all the essencial changes needed to correct problems in the TTCN implementation of test case 8.1.4.2.1.1 which is part of the NR5GC test suite in the ‘iwd-TTCN3-B2019-06_D19wk38’ ATS delivery. 

With these changes applied the test case can be demonstrated to run with one or more UEs (see section 4). Execution log files are provided as evidence.

3. Verification Test Summary

Test Case:
8.1.4.2.1.1
Test Group:
\NR5GC\8_1_4\
ATS Version:
iwd-TTCN3-B2019-06_D19wk38
System Simulator used:
Keysight 5G Protocol Conformance Toolset
UE used:
HiSilicon Balong 5000
Verification Status:
PASS


4. Corrections required for NR5GC test case 8.1.4.2.1.1 on wk38 ATS

Changes 1-5 from R5s191107 are also required and applied.
4.1 Change 1

	Function name
	f_EUTRA_TrackingAreaUpdate_Accept_Complete()

	Reason for change
	Following handover from 5GNR -> EUTRA, SRB2 will be active and so will be used for the Tracking Area Update signalling

	Summary of change
	Add a parameter to allow use of SRB 2

	TTCN module
	EUTRA_NASSteps.ttcn

	MCC160 Comment
	


Before change

  function f_EUTRA_TrackingAreaUpdate_Accept_Complete(EUTRA_CellId_Type                             p_CellId,

                                                      GutiParameters_Type                     p_GutiParams,

                                                      B3_Type                                 p_EpsUpdate_TypeValue,

                                                      template (omit) EPS_BearerContextStatus p_EPS_ContextStatus,

                                                      template (omit) AdditionalUpdateType    p_AdditionalUpdateType, // @sic R5s170351 sic@

                                                      EUTRA_ATTACH_Type                       p_ForcedAttach,

                                                      template (omit) PLMN_List               p_EPLMNlist := omit,

                                                      template (omit) EPS_NetworkFeatureSupport p_NwkFeatSup := cs_EPS_NwkFtSup_IMSVoice,

                                                      template (omit) GPRS_Timer p_T3412 := cs_GprsTimer_tv('5A'O, tsc_GprsTimerUnit_deact, '00000'B)) runs on EUTRA_PTC

  {

    var NAS_PlmnId       v_PLMN := f_Asn2Nas_PlmnId(p_GutiParams.PLMN_Identity);

    var NAS_Lac          v_LAC := f_EUTRA_CellInfo_GetLocationAreaCode(p_CellId);

    var TrackingAreaCode v_Tac := f_EUTRA_CellInfo_GetTAC(p_CellId);

    var template (omit)  LocAreaId               v_LAI := f_GetLAI(v_PLMN, v_LAC, p_ForcedAttach); // @sic R5s110176 sic@

    var template (value) TrackingAreaIdList      v_TaiList := cds_TAIListNonConsecutive_tlv(v_PLMN, { bit2oct(v_Tac) }); // @sic R5s100029 sic@

    var template (omit)  MobileIdentity          v_MSId := f_GetMSId(p_ForcedAttach); // @sic R5-103673 sic@

    //    var template (omit)  EPS_BearerContextStatus v_EPS_ContextStatus; // @sic R5s100608 sic@

    /* Transmit TRACKING AREA UPDATE ACCEPT message */

    /* In case MME has the cached security context it verifies the TAU Request message

       and replies with TAU Accept message protected with the cached security context

       ( 33.401 cl. 9.1.2 Idle mode procedures in E-UTRAN ) */

    SRB.send(cas_SRB1_NasPdu_REQ(p_CellId,

                                 cs_TimingInfo_Now,

                                 cs_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered,

                                                cs_TAU_Accept_Common(p_EpsUpdate_TypeValue,

                                                                     f_GutiParameters2MobileIdentity (tsc_IEI_Guti, p_GutiParams),

                                                                     v_TaiList,

                                                                     p_EPS_ContextStatus,

                                                                     v_LAI,

                                                                     v_MSId,

                                                                     f_GetAdditionalUpdateResult(p_AdditionalUpdateType, p_ForcedAttach), // @sic R5s110176 sic@

                                                                     -,

                                                                     -,

                                                                     p_EPLMNlist,

                                                                     p_NwkFeatSup, // @sic R5-110771 sic@

                                                                     p_T3412)))); // @sic R5-180728 sic@

    /* Receive TRACKING AREA UPDATE COMPLETE */

    /* According to 24.301 cl: 5.5.3.2.4:

       If the TRACKING AREA UPDATE ACCEPT message contained a GUTI,

       the UE shall return a TRACKING AREA UPDATE COMPLETE message to

       the MME to acknowledge the received GUTI */

    SRB.receive(car_SRB1_NasPdu_IND(p_CellId,

                                    cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,

                                                            cr_508_TAU_Complete)));

  }
After change

function f_EUTRA_TrackingAreaUpdate_Accept_Complete(EUTRA_CellId_Type                             p_CellId,

                                                      GutiParameters_Type                     p_GutiParams,

                                                      B3_Type                                 p_EpsUpdate_TypeValue,

                                                      template (omit) EPS_BearerContextStatus p_EPS_ContextStatus,

                                                      template (omit) AdditionalUpdateType    p_AdditionalUpdateType, // @sic R5s170351 sic@

                                                      EUTRA_ATTACH_Type                       p_ForcedAttach,

                                                      template (omit) PLMN_List               p_EPLMNlist := omit,

                                                      template (omit) EPS_NetworkFeatureSupport p_NwkFeatSup := cs_EPS_NwkFtSup_IMSVoice,

                                                      template (omit) GPRS_Timer p_T3412 := cs_GprsTimer_tv('5A'O, tsc_GprsTimerUnit_deact, '00000'B, boolean p_UseSRB2 := false)) runs on EUTRA_PTC

  {

    var NAS_PlmnId       v_PLMN := f_Asn2Nas_PlmnId(p_GutiParams.PLMN_Identity);

    var NAS_Lac          v_LAC := f_EUTRA_CellInfo_GetLocationAreaCode(p_CellId);

    var TrackingAreaCode v_Tac := f_EUTRA_CellInfo_GetTAC(p_CellId);

    var template (omit)  LocAreaId               v_LAI := f_GetLAI(v_PLMN, v_LAC, p_ForcedAttach); // @sic R5s110176 sic@

    var template (value) TrackingAreaIdList      v_TaiList := cds_TAIListNonConsecutive_tlv(v_PLMN, { bit2oct(v_Tac) }); // @sic R5s100029 sic@

    var template (omit)  MobileIdentity          v_MSId := f_GetMSId(p_ForcedAttach); // @sic R5-103673 sic@

    //    var template (omit)  EPS_BearerContextStatus v_EPS_ContextStatus; // @sic R5s100608 sic@

    /* Transmit TRACKING AREA UPDATE ACCEPT message */

    /* In case MME has the cached security context it verifies the TAU Request message

       and replies with TAU Accept message protected with the cached security context

       ( 33.401 cl. 9.1.2 Idle mode procedures in E-UTRAN ) */
if (p_UseSRB2)

{

    SRB.send(cas_SRB2_NasPdu_REQ(p_CellId,

                                 cs_TimingInfo_Now,

                                 cs_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered,

                                                cs_TAU_Accept_Common(p_EpsUpdate_TypeValue,

                                                                     f_GutiParameters2MobileIdentity (tsc_IEI_Guti, p_GutiParams),

                                                                     v_TaiList,

                                                                     p_EPS_ContextStatus,

                                                                     v_LAI,

                                                                     v_MSId,

                                                                     f_GetAdditionalUpdateResult(p_AdditionalUpdateType, p_ForcedAttach), // @sic R5s110176 sic@

                                                                     -,

                                                                     -,

                                                                     p_EPLMNlist,

                                                                     p_NwkFeatSup, // @sic R5-110771 sic@

                                                                     p_T3412)))); // @sic R5-180728 sic@

}

else

{
    SRB.send(cas_SRB1_NasPdu_REQ(p_CellId,

                                 cs_TimingInfo_Now,

                                 cs_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered,

                                                cs_TAU_Accept_Common(p_EpsUpdate_TypeValue,

                                                                     f_GutiParameters2MobileIdentity (tsc_IEI_Guti, p_GutiParams),

                                                                     v_TaiList,

                                                                     p_EPS_ContextStatus,

                                                                     v_LAI,

                                                                     v_MSId,

                                                                     f_GetAdditionalUpdateResult(p_AdditionalUpdateType, p_ForcedAttach), // @sic R5s110176 sic@

                                                                     -,

                                                                     -,

                                                                     p_EPLMNlist,

                                                                     p_NwkFeatSup, // @sic R5-110771 sic@

                                                                     p_T3412)))); // @sic R5-180728 sic@

 }   

    /* Receive TRACKING AREA UPDATE COMPLETE */

    /* According to 24.301 cl: 5.5.3.2.4:

       If the TRACKING AREA UPDATE ACCEPT message contained a GUTI,

       the UE shall return a TRACKING AREA UPDATE COMPLETE message to

       the MME to acknowledge the received GUTI */
    if (p_UseSrb2)

    {

         SRB.receive(car_SRB2_NasPdu_IND(p_CellId,

                                    cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,

                                                            cr_508_TAU_Complete)));

    }

    else

    {
         SRB.receive(car_SRB1_NasPdu_IND(p_CellId,

                                    cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,

                                                            cr_508_TAU_Complete)));

    }
  }
4.2 Change 2
	Function name
	f_NG_Authentication_A14()

	Reason for change
	Incorrect FC is being used, it should be 0x73 for mobility registration and 0x74 for handover according 33.501 Annex A.14

	Summary of change
	Correct logic in function

	TTCN module
	EUTRA_NR_SecurityFunctions

	MCC160 Comment
	


Before change

  function f_NG_Authentication_A14(boolean p_Handover,

                                  O4_Type p_Count,

                                  B256_Type p_KAMF,

                                  KDF_Type p_KDF_Type) return B256_Type

  {

    const octetstring const_S73_FC :='73'O;

    const octetstring const_S74_FC :='74'O;

    var octetstring v_S;

    if (p_Handover) {

      // Generation of String

      v_S := const_S73_FC;

      //FC = 0x73

    } else { // mobility registration

      // Generation of String

      v_S := const_S74_FC;

      //FC = 0x74

    }

    v_S :=  (v_S & p_Count);

    //P0 = Count

    v_S :=  (v_S & '0004'O) ;

    //L0 = length of count

    return fx_KeyDerivationFunction(p_KDF_Type, p_KAMF, v_S);

  };

After change

  function f_NG_Authentication_A14(boolean p_Handover,

                                  O4_Type p_Count,

                                  B256_Type p_KAMF,

                                  KDF_Type p_KDF_Type) return B256_Type

  {

    const octetstring const_S73_FC :='73'O;

    const octetstring const_S74_FC :='74'O;

    var octetstring v_S;

    if (not p_Handover) {

      // Generation of String

      v_S := const_S73_FC;

      //FC = 0x73

    } else { // mobility registration

      // Generation of String

      v_S := const_S74_FC;

      //FC = 0x74

    }

    v_S :=  (v_S & p_Count);

    //P0 = Count

    v_S :=  (v_S & '0004'O) ;

    //L0 = length of count

    return fx_KeyDerivationFunction(p_KDF_Type, p_KAMF, v_S);

  };
4.3 Change 3
	Function name
	f_Get_NG_SecurityModeCmdMsg()

	Reason for change
	The selected EPS security algorithms should be provided to the UE prior to 5GC->EPC handover. In the current implementation this is given in the SECURITY MODE COMMAND only if the UE has indicated support of S1 mode in GMMCapabilityInfo in REGISTRATION REQUEST. Since this is only sent to the UE after NAS security is started (non-cleartext IE), the EPS security algorithms are currently never provided to the UE.

	Summary of change
	Always add selected EPS security algorithms to SECURITY MODE COMMAND. 

Note : 24.501 5.4.2.1 states that the NW can initiate a security mode procedure in order to provide the selected EPS NAS security algorithms to the UE. So the alternative more realistic solution would be to modify the preamble sequence to send a 2nd SECURITY MODE COMMAND after the 1st SECURITY MODE COMPLETE is recieved if the embedded full REGISTRATION REQUEST indicates S1 support.

	TTCN module
	EUTRA38_NR5GCProcedures

	MCC160 Comment
	


Before change

 function f_Get_NG_SecurityModeCmdMsg(NG_NAS_SecurityParams_Type p_SecurityParams,

                                       TypeOfRegistration_Type p_RegType,

                                       GMM_MobilityInfo_Type p_GMM_MobilityInfo,

                                       boolean p_IMEISVReq := false,

                                       template (omit) AdditionalSecurityInfo p_AddInfo := omit,

                                       template (omit) EAP_Message p_EAP := omit,

                                       template (omit) ABBA p_ABBA := omit) return template (value) NG_NAS_DL_Message_Type

  {

    var template (value) NG_NAS_DL_Message_Type v_NASMsg;

    var template (omit)  AdditionalSecurityInfo v_AddInfo := p_AddInfo;

    var template (omit) IMEISV_Request v_IMEISV_Request := omit;

    var template (omit) NAS_SecurityAlgorithms v_EPSAlgs := omit;

    var NG_UE_SecurityCapability v_NG_UE_SecurityCapability;

    var template (omit) S1_UE_SecurityCapability v_ReplayedCap := omit;

    if (isvalue(p_GMM_MobilityInfo.UECap)) { // @sic R5s190109 sic@

      v_NG_UE_SecurityCapability := p_GMM_MobilityInfo.UECap;

      v_NG_UE_SecurityCapability.iei := omit;

    } else {

      FatalError(__FILE__, __LINE__, "UE Capabilities not included in Registration Request");

    }

    if (isvalue(p_GMM_MobilityInfo.GMMCap)) {

      if (match('1'B, p_GMM_MobilityInfo.GMMCap.s1Cap)) {

        v_EPSAlgs := cs_NAS_SecurityAlgorithms(px_NAS_CipheringAlgorithm, px_NAS_IntegrityProtAlgorithm, '57'O); // @sic R5s190285 sic@
        if (isvalue(p_GMM_MobilityInfo.S1NwkCap)) {

          v_ReplayedCap.ieValue := f_EPS_FillNAS_SecurityCapability_Common(p_GMM_MobilityInfo.S1NwkCap, omit);  // @sic R5s190421 sic@

          v_ReplayedCap.iei := '19'O; // @sic R5s190421 sic@

        }

      }

    }

    if (p_IMEISVReq) {

      v_IMEISV_Request := cds_NG_IMEISV_Request;

    }

    if (p_RegType == Initial_NoSecurity) {

    // Must ensure UE sends complete initial message if SS doesn't have security

      if (isvalue(v_AddInfo)) {

        v_AddInfo.rinmr := '1'B;

      } else {

        v_AddInfo := cs_AdditionalSecurityInfo('1'B);

      }

    }

    v_NASMsg := cs_NG_SECURITY_MODE_COMMAND(cs_NG_NAS_SecurityAlgorithms (p_SecurityParams.NAS_Ciphering.Algorithm, p_SecurityParams.NAS_Integrity.Algorithm),

                                            p_SecurityParams.KSIamf,

                                            v_NG_UE_SecurityCapability,

                                            v_IMEISV_Request,

                                            v_EPSAlgs,

                                            v_AddInfo,

                                            p_EAP,

                                            p_ABBA,

                                            v_ReplayedCap);

    return v_NASMsg;

  }
After change

  function f_Get_NG_SecurityModeCmdMsg(NG_NAS_SecurityParams_Type p_SecurityParams,

                                       TypeOfRegistration_Type p_RegType,

                                       GMM_MobilityInfo_Type p_GMM_MobilityInfo,

                                       boolean p_IMEISVReq := false,

                                       template (omit) AdditionalSecurityInfo p_AddInfo := omit,

                                       template (omit) EAP_Message p_EAP := omit,

                                       template (omit) ABBA p_ABBA := omit) return template (value) NG_NAS_DL_Message_Type

  {

    var template (value) NG_NAS_DL_Message_Type v_NASMsg;

    var template (omit)  AdditionalSecurityInfo v_AddInfo := p_AddInfo;

    var template (omit) IMEISV_Request v_IMEISV_Request := omit;

    var template (omit) NAS_SecurityAlgorithms v_EPSAlgs := omit;

    var NG_UE_SecurityCapability v_NG_UE_SecurityCapability;

    var template (omit) S1_UE_SecurityCapability v_ReplayedCap := omit;

    if (isvalue(p_GMM_MobilityInfo.UECap)) { // @sic R5s190109 sic@

      v_NG_UE_SecurityCapability := p_GMM_MobilityInfo.UECap;

      v_NG_UE_SecurityCapability.iei := omit;

    } else {

      FatalError(__FILE__, __LINE__, "UE Capabilities not included in Registration Request");

    }

    v_EPSAlgs := cs_NAS_SecurityAlgorithms(px_NAS_CipheringAlgorithm, px_NAS_IntegrityProtAlgorithm, '57'O); // @sic R5s190285 sic@
     if (isvalue(p_GMM_MobilityInfo.GMMCap)) {

     //REMOVED  if (match('1'B, p_GMM_MobilityInfo.GMMCap.s1Cap)) {

       //MOVED  v_EPSAlgs := cs_NAS_SecurityAlgorithms(px_NAS_CipheringAlgorithm, px_NAS_IntegrityProtAlgorithm, '57'O); // @sic R5s190285 sic@
        if (isvalue(p_GMM_MobilityInfo.S1NwkCap)) {

          v_ReplayedCap.ieValue := f_EPS_FillNAS_SecurityCapability_Common(p_GMM_MobilityInfo.S1NwkCap, omit);  // @sic R5s190421 sic@

          v_ReplayedCap.iei := '19'O; // @sic R5s190421 sic@

        }

      //}
    }

    if (p_IMEISVReq) {

      v_IMEISV_Request := cds_NG_IMEISV_Request;

    }

    if (p_RegType == Initial_NoSecurity) {

    // Must ensure UE sends complete initial message if SS doesn't have security

      if (isvalue(v_AddInfo)) {

        v_AddInfo.rinmr := '1'B;

      } else {

        v_AddInfo := cs_AdditionalSecurityInfo('1'B);

      }

    }

    v_NASMsg := cs_NG_SECURITY_MODE_COMMAND(cs_NG_NAS_SecurityAlgorithms (p_SecurityParams.NAS_Ciphering.Algorithm, p_SecurityParams.NAS_Integrity.Algorithm),

                                            p_SecurityParams.KSIamf,

                                            v_NG_UE_SecurityCapability,

                                            v_IMEISV_Request,

                                            v_EPSAlgs,

                                            v_AddInfo,

                                            p_EAP,

                                            p_ABBA,

                                            v_ReplayedCap);

    return v_NASMsg;

  }
4.4 Change 4
	Function name
	Template cs_SecurityConfigHO_From5GToEPC()

	Reason for change
	NextHopChainingCount for handover from 5G to EPC should be 2 according to 33.501 8.3.2

	Summary of change
	Change nextHopChainingCount to 2

Note : A prose CR on 38.523-1 will be raised at RAN5#86

	TTCN module
	IRAT_Handover_NR5GC_EUTRA

	MCC160 Comment
	


Before change

  template (value) SecurityConfigHO_v1530 cs_SecurityConfigHO_From5GToEPC(IntegrityProtAlgorithm_Type    p_AS_IntegrityProtAlgorithm := px_RRC_IntegrityProtAlgorithm,

                                                                 CipheringAlgorithm_r12         p_AS_CipheringAlgorithm := px_RRC_CipheringAlgorithm) :=

  {

    handoverType_v1530  := {

      fivegc_ToEPC_r15  := {

        securityAlgorithmConfig_r15 := {

          cipheringAlgorithm := p_AS_CipheringAlgorithm,

          integrityProtAlgorithm := p_AS_IntegrityProtAlgorithm

        },

          nextHopChainingCount_r15  := 0
      }

    }

  };

After change

  template (value) SecurityConfigHO_v1530 cs_SecurityConfigHO_From5GToEPC(IntegrityProtAlgorithm_Type    p_AS_IntegrityProtAlgorithm := px_RRC_IntegrityProtAlgorithm,

                                                                 CipheringAlgorithm_r12         p_AS_CipheringAlgorithm := px_RRC_CipheringAlgorithm) :=

  {

    handoverType_v1530  := {

      fivegc_ToEPC_r15  := {

        securityAlgorithmConfig_r15 := {

          cipheringAlgorithm := p_AS_CipheringAlgorithm,

          integrityProtAlgorithm := p_AS_IntegrityProtAlgorithm

        },

        nextHopChainingCount_r15  := 2
      }

    }

  };
4.5 Change 5
	Function name
	f_Encode_RRCConnectionReconfiguration_HO5GToEUTRA()

	Reason for change
	for EUTRA FDD, both DL and UL frequency need to be populated 



	Summary of change
	Replace ARFN parameter with CarrierFreqEUTRA

	TTCN module
	IRAT_Handover_NR5GC_EUTRA

	MCC160 Comment
	


Before change

 function f_Encode_RRCConnectionReconfiguration_HO5GToEUTRA(integer p_Arfcn,

                                                             Dl_Bandwidth_Type p_DL_ChBandwidth,

                                                             Ul_Bandwidth_Type p_UL_ChBandwidth,

                                                             integer p_PhysCellId,

                                                             integer p_RootSequenceIndex,

                                                             integer p_N := 1) return octetstring

  {

    var bitstring v_Bitstring;

    var ChannelBandwidthDependency_Type v_ChannelBandwidthDependency := f_EUTRA_BandDependentParam(p_DL_ChBandwidth,

                                                                                                   p_UL_ChBandwidth);

    var template (value) DRB_ToAddModList v_DRB_ToAddModList := { cs_508_DRB_ToAddMod_DEFAULT_AM(tsc_DRB1) }

    var template (omit) CarrierFreqEUTRA v_CarrierFreqEUTRA := omit;

    var template (omit) CarrierFreqEUTRA_v9e0 v_CarrierFreqEUTRA_v9e0 := omit;

    var template (value) RRCConnectionReconfiguration_v1310_IEs v_RRCConnectionReconfiguration_v1310_IEs := cs_RRCConnectionReconfiguration_v1310_IEs (cs_RRCConnectionReconfiguration_v1430_IEs (-,

                                                                                                                     cs_RRCConnectionReconfiguration_v1510_IEs (omit, -, -, -,
                                                                                                                       cs_RRCConnectionReconfiguration_v1530_IEs (cs_SecurityConfigHO_From5GToEPC()))));

    var template (value) RRCConnectionReconfiguration_v890_IEs v_RRCConnectionReconfiguration_v890_IEs := cs_RRCConnectionReconfiguration_v890_IEs(cs_RRCConnectionReconfiguration_v920_IEs(-, -,

                                                                                                                       cs_RRCConnectionReconfiguration_v1020_IEs (-, -,

                                                                                                                         cs_RRCConnectionReconfiguration_v1130_IEs (-,

                                                                                                                           cs_RRCConnectionReconfiguration_v1250_IEs (-, -, -, v_RRCConnectionReconfiguration_v1310_IEs)))));

    if (p_Arfcn <= maxEARFCN) { //Band < 64

        v_CarrierFreqEUTRA := cs_CarrierFreqEUTRA(p_Arfcn, omit);

    }

    if (p_Arfcn >= maxEARFCN) { //Band > 64

        v_CarrierFreqEUTRA_v9e0 := cs_CarrierFreqEUTRA_v9e0(p_Arfcn, omit);

    }
    ………………….
    return bit2oct(f_OctetAlignedBitString(v_Bitstring));  // NOTE: bit2oct would add additional bits at the beginning (MSB) whereas here it has be added at the end (LSB)

  }  
After change

   //KS R5sxxxx

    function f_Encode_RRCConnectionReconfiguration_HO5GToEUTRA(                                                            CellCarrierFreqEUTRA_Type p_CarrierFreqEUTRA,
                                                             Dl_Bandwidth_Type p_DL_ChBandwidth,

                                                             Ul_Bandwidth_Type p_UL_ChBandwidth,

                                                             integer p_PhysCellId,

                                                             integer p_RootSequenceIndex,

                                                             integer p_N := 1) return octetstring

  {

    var bitstring v_Bitstring;

    var ChannelBandwidthDependency_Type v_ChannelBandwidthDependency := f_EUTRA_BandDependentParam(p_DL_ChBandwidth,

                                                                                                   p_UL_ChBandwidth);

    var template (value) DRB_ToAddModList v_DRB_ToAddModList := { cs_508_DRB_ToAddMod_DEFAULT_AM(tsc_DRB1) }

    var template (omit) CarrierFreqEUTRA v_CarrierFreqEUTRA := omit;

    var template (omit) CarrierFreqEUTRA_v9e0 v_CarrierFreqEUTRA_v9e0 := omit;

    var template (value) RRCConnectionReconfiguration_v1310_IEs v_RRCConnectionReconfiguration_v1310_IEs := cs_RRCConnectionReconfiguration_v1310_IEs (cs_RRCConnectionReconfiguration_v1430_IEs (-,

                                                                                                                     cs_RRCConnectionReconfiguration_v1510_IEs (omit, -, -, -, 
                                                                                                                                                                                                                                                  cs_RRCConnectionReconfiguration_v1530_IEs (cs_SecurityConfigHO_From5GToEPC()))));                                                                                                             

    var template (value) RRCConnectionReconfiguration_v890_IEs v_RRCConnectionReconfiguration_v890_IEs := cs_RRCConnectionReconfiguration_v890_IEs(cs_RRCConnectionReconfiguration_v920_IEs(-, -,

                                                                                                                       cs_RRCConnectionReconfiguration_v1020_IEs (-, -,

                                                                                                                         cs_RRCConnectionReconfiguration_v1130_IEs (-,

                                                                                                                           cs_RRCConnectionReconfiguration_v1250_IEs (-, -, -, v_RRCConnectionReconfiguration_v1310_IEs)))));

    if (p_CarrierFreqEUTRA.dl_CarrierFreq <= maxEARFCN) { //Band < 64

        v_CarrierFreqEUTRA := cs_CarrierFreqEUTRA(p_CarrierFreqEUTRA.dl_CarrierFreq, p_CarrierFreqEUTRA.ul_CarrierFreq);

    }

    if (p_CarrierFreqEUTRA.dl_CarrierFreq >= maxEARFCN) { //Band > 64

        v_CarrierFreqEUTRA_v9e0 := cs_CarrierFreqEUTRA_v9e0(p_CarrierFreqEUTRA.dl_CarrierFreq, omit);

    }

    ………………
  }
4.6 Change 6
	Function name
	fl_TC_8_1_4_2_1_1_NR5GC_EUTRA_TestBody()

	Reason for change
	1. Once RRCConnectionReconfiguration is sent to NR it is immediately sent to the UE in MobilityFromNRCommand, so there can be a race condition between the SS configuration and receving the RRCConnectionRecofigurationComplete

2.  Random access procedure can be triggered because periodic TA is not started on EUTRA cell

3. Following handover from 5GNR -> EUTRA, SRB2 will be active and so will be used for the Tracking Area Update signalling
4. Depending on UE capabilities, the Update Type in TRACKING AREA UPDATE REQUEST may be “TA/LA Updating with IMSI attach”

	Summary of change
	1. Send RRCConnectionReconfiguration to NR after local SS configuration is done 
2. Start periodic TA on EUTRA cell
3. Receive TRACKING AREA UPDATE REQUEST on SRB2

4. Extend expected attach type

	TTCN module
	IRAT_Handover_NR5GC_EUTRA

	MCC160 Comment
	


Before change

  function fl_TC_8_1_4_2_1_1_NR5GC_EUTRA_TestBody() runs on EUTRA_5GS_PTC

  {

    var EUTRA_NR_CoOrd_GUTI_Type v_CoOrd_Guti;

    var EUTRA_NR_CoOrd_Security_Type v_SecurityRx;

    var template (value) MobileIdentity v_MappedGUTI;

    var SRB_COMMON_IND v_ReceivedMsg;

    var template AdditionalUpdateType v_AdditionalUpdateType := f_GetAdditionalUpdateType (NORMAL);

    var EUTRA_SecurityParams_Type v_Auth_Params := f_EUTRA_Security_Get();

    var octetstring v_HOCmd := f_Encode_RRCConnectionReconfiguration_HO5GToEUTRA(f_EUTRA_CellInfo_GetEARFCN_DL_r9(eutra_Cell1),

                                                                                 f_EUTRA_CellInfo_GetDL_ChBandwidth (eutra_Cell1),

                                                                                 f_EUTRA_CellInfo_GetUL_ChBandwidth (eutra_Cell1),

                                                                                 f_EUTRA_CellInfo_GetPhyCellId(eutra_Cell1),

                                                                                 f_EUTRA_CellInfo_GetRootSequenceIndex (eutra_Cell1));

    // Send EUTRA HO command to NR

    f_EUTRA_NR_SendCoOrd(NR, cms_EUTRA_NR_OctetData (v_HOCmd, "HO Cmd"));

    v_CoOrd_Guti := f_EUTRA_NR_WaitForCoOrd_GutiParams(NR);

    v_SecurityRx := f_EUTRA_NR_WaitForCoOrd_Security (NR);

    v_MappedGUTI := f_EUTRA38_NR_Convert5G_CoOrdToEPS_GUTI (v_CoOrd_Guti, omit);

    // Configure C-RNTI based contention resolution in target cell

    f_EUTRA_SS_ConfigRachProcedure(eutra_Cell1, tsc_C_RNTI_Def2, cs_RachProcedureConfig_CRNTI_Dedicated(f_EUTRA_CellInfo_GetRAR_TA(eutra_Cell1),

                                                                                                        f_EUTRA_CellInfo_GetDL_ChBandwidth(eutra_Cell1))); // @sic R5-123712 sic@

    f_EUTRA_CellInfo_SetC_RNTI ( eutra_Cell1, tsc_C_RNTI_Def2 );

    // activate RRC security at SS target cell for SRB 1, 2 and 1 AM DRB

    v_Auth_Params := f_EUTRA38_MappedContextFromN1_InitNAS(eutra_Cell1, v_SecurityRx, true);

    v_Auth_Params.AS_Ciphering.ActTimeList := f_EUTRA_RRC_CipherActTime_GetForHO(eutra_Cell1, tsc_DrbList_Def);

    f_EUTRA_NB_SS_NAS_ConfigureSecurity(NASCTRL, v_Auth_Params.NAS_Integrity, v_Auth_Params.NAS_Ciphering);

    f_EUTRA_NB_SS_NAS_CountSet(NASCTRL, '00000000'O, omit);

    f_EUTRA_SS_RRC_EnableIntProt_CiphULandDL(eutra_Cell1,

                                             v_Auth_Params.AS_Integrity,

                                             v_Auth_Params.AS_Ciphering,

                                             cs_TimingInfo_Now);

    f_EUTRA_Security_Set(v_Auth_Params);

    //Configure UL grant configuration ("OnSR", periodic TA is NOT started) in target cell (cell 1)

    f_EUTRA_SS_CommonCellConfig(eutra_Cell1, cas_ULGrantAllocation_Def_REQ(eutra_Cell1, cs_TimingInfo_Now, cs_UplinkTimeAlignment_Synch_None));

    // @siclog "Step 2" siclog@

    SRB.receive( car_SRB1_RrcPdu_IND( eutra_Cell1, cr_508_RRCConnectionReconfigurationComplete( tsc_RRC_TI_Def ) ) );

    /* SS automatically checks that UE transmit an RRCConnectionReconfigurationComplete using the indicated

       security key and the indicated security algorithms */

    f_EUTRA_PreliminaryPass( __FILE__, __LINE__, "Step 2" );

    SRB.receive(car_SRB1_NasPdu_IND(eutra_Cell1,

                                    cr_NAS_Indication(tsc_SHT_IntegrityProtected,

                                                      cdr_TAU_Request_N1ToS1(tsc_EpsUpdate_TaUpdate,

                                                                            v_SecurityRx.KSI,

                                                                            cr_DRXparameter_Any ('5C'O) ifpresent,

                                                                            v_AdditionalUpdateType,

                                                                            v_MappedGUTI,

                                                                            cr_TAI_Any ifpresent))))-> value v_ReceivedMsg;

    f_EUTRA38_TrackingAreaUpdateFromN1_WithoutReq (eutra_Cell1,

                                                   tsc_RRC_TI_Def,

                                                   v_ReceivedMsg.Signalling.Nas[0],

                                                   -, noRrcConnectionRelease);

  } 
After change

function fl_TC_8_1_4_2_1_1_NR5GC_EUTRA_TestBody() runs on EUTRA_5GS_PTC

  {

    var EUTRA_NR_CoOrd_GUTI_Type v_CoOrd_Guti;

    var EUTRA_NR_CoOrd_Security_Type v_SecurityRx;

    var template (value) MobileIdentity v_MappedGUTI;

    var SRB_COMMON_IND v_ReceivedMsg;

    var template AdditionalUpdateType v_AdditionalUpdateType := f_GetAdditionalUpdateType (NORMAL);

    var EUTRA_SecurityParams_Type v_Auth_Params := f_EUTRA_Security_Get();

var octetstring v_HOCmd := f_Encode_RRCConnectionReconfiguration_HO5GToEUTRA(f_EUTRA_CellInfo_GetEARFCN_DL_r9(eutra_Cell1),

                                                                                 f_EUTRA_CellInfo_GetDL_ChBandwidth (eutra_Cell1),

                                                                                 f_EUTRA_CellInfo_GetUL_ChBandwidth (eutra_Cell1),

                                                                                 f_EUTRA_CellInfo_GetPhyCellId(eutra_Cell1),

                                                                                 f_EUTRA_CellInfo_GetRootSequenceIndex (eutra_Cell1));

          // Send EUTRA HO command to NR

    //MOVED f_EUTRA_NR_SendCoOrd(NR, cms_EUTRA_NR_OctetData (v_HOCmd, "HO Cmd"));

    v_CoOrd_Guti := f_EUTRA_NR_WaitForCoOrd_GutiParams(NR);

    v_SecurityRx := f_EUTRA_NR_WaitForCoOrd_Security (NR);

    v_MappedGUTI := f_EUTRA38_NR_Convert5G_CoOrdToEPS_GUTI (v_CoOrd_Guti, omit);

    // Configure C-RNTI based contention resolution in target cell

    f_EUTRA_SS_ConfigRachProcedure(eutra_Cell1, tsc_C_RNTI_Def2, cs_RachProcedureConfig_CRNTI_Dedicated(f_EUTRA_CellInfo_GetRAR_TA(eutra_Cell1),

                                                                                                        f_EUTRA_CellInfo_GetDL_ChBandwidth(eutra_Cell1))); // @sic R5-123712 sic@

    f_EUTRA_CellInfo_SetC_RNTI ( eutra_Cell1, tsc_C_RNTI_Def2 );

    // activate RRC security at SS target cell for SRB 1, 2 and 1 AM DRB

    v_Auth_Params := f_EUTRA38_MappedContextFromN1_InitNAS(eutra_Cell1, v_SecurityRx, true);

    v_Auth_Params.AS_Ciphering.ActTimeList := f_EUTRA_RRC_CipherActTime_GetForHO(eutra_Cell1, tsc_DrbList_Def);

    f_EUTRA_NB_SS_NAS_ConfigureSecurity(NASCTRL, v_Auth_Params.NAS_Integrity, v_Auth_Params.NAS_Ciphering);

    f_EUTRA_NB_SS_NAS_CountSet(NASCTRL, '00000000'O, omit);

    f_EUTRA_SS_RRC_EnableIntProt_CiphULandDL(eutra_Cell1,

                                             v_Auth_Params.AS_Integrity,

                                             v_Auth_Params.AS_Ciphering,

                                             cs_TimingInfo_Now);

    f_EUTRA_Security_Set(v_Auth_Params);

    //Configure UL grant configuration ("OnSR", periodic TA is NOT started) in target cell (cell 1)

    //KS R5sxxxx
    f_EUTRA_SS_CommonCellConfig(eutra_Cell1, cas_ULGrantAllocation_Def_REQ(eutra_Cell1, cs_TimingInfo_Now, cs_UplinkTimeAlignment_Synch_Auto_Def));

    f_EUTRA_NR_SendCoOrd(NR, cms_EUTRA_NR_OctetData (v_HOCmd, "HO Cmd"));

    // @siclog "Step 2" siclog@

    SRB.receive( car_SRB1_RrcPdu_IND( eutra_Cell1, cr_508_RRCConnectionReconfigurationComplete( tsc_RRC_TI_Def ) ) );

    /* SS automatically checks that UE transmit an RRCConnectionReconfigurationComplete using the indicated

       security key and the indicated security algorithms */

    f_EUTRA_PreliminaryPass( __FILE__, __LINE__, "Step 2" );

    SRB.receive(car_SRB2_NasPdu_IND(eutra_Cell1,

                                    cr_NAS_Indication(tsc_SHT_IntegrityProtected,

                                                      cdr_TAU_Request_N1ToS1((tsc_EpsUpdate_Combined_IMSIAttach, tsc_EpsUpdate_TaUpdate ),
                                                                            v_SecurityRx.KSI,

                                                                            cr_DRXparameter_Any ('5C'O) ifpresent,

                                                                            v_AdditionalUpdateType,

                                                                            v_MappedGUTI,

                                                                            cr_TAI_Any ifpresent))))-> value v_ReceivedMsg;
    f_EUTRA38_TrackingAreaUpdateFromN1_WithoutReq (eutra_Cell1,

                                                   tsc_RRC_TI_Def,

                                                   v_ReceivedMsg.Signalling.Nas[0],

                                                   -, noRrcConnectionRelease);

  }
4.7 Change 7 
	Function name
	f_TC_8_1_4_2_1_1_NR5GC_EUTRA()

	Reason for change
	Following 5GC->EPC handover, the TRACKING AREA UPDATE ACCEPT at step 4 will indicate “TA Updated” and the UE will consider itself attached for EPS services only. Due to this the UE will indicate “EPS Detach” on switch off, regardless of its support for combined attach procedures.

	Summary of change
	Pass the parameter “p_ForcedAttach” as EPS-Only to the postamble function so that “EPS Detach” is expected.
Note : Alternative solution is to set the EPS Update Result based at step 4 based on the received EPS Update Type at step 3 (see change 6.4), then the existing checking for detach type will be correct without forcing.

	TTCN module
	IRAT_Handover_NR5GC_EUTRA

	MCC160 Comment
	


Before change

   function f_TC_8_1_4_2_1_1_NR5GC_EUTRA() runs on EUTRA_5GS_PTC

  { // Inter-RAT Handover / From NR to E-UTRA / Success

    f_EUTRA38_IRAT_Init(c1); // FFS do we need c31 is to be applied?

    //Create and configure cell

    f_EUTRA_CellConfig_Def(eutra_Cell1);

    // FFS - do we need to attach on EUTRA too?

    f_EUTRA_NR_WaitForCoOrd_Trigger (NR, "Preamble");

    // FFS Prose doesn't specify any power levels

    f_EUTRA38_SetCellPower(eutra_Cell1, tsc_ServingCellRS_EPRE, tsc_ServingCellRS_EPRE); //  FFS

    f_EUTRA_TestBody_Set(true);

    fl_TC_8_1_4_2_1_1_NR5GC_EUTRA_TestBody ();

    f_EUTRA_TestBody_Set(false);

    f_EUTRA_Postamble(eutra_Cell1, E2_CONNECTED);

    // take down the NR cell

    f_EUTRA_NR_SendCoOrd (NR, cms_EUTRA_NR_Trigger("Postamble"));

  }
After change

function f_TC_8_1_4_2_1_1_NR5GC_EUTRA() runs on EUTRA_5GS_PTC

  { // Inter-RAT Handover / From NR to E-UTRA / Success

    f_EUTRA38_IRAT_Init(c1); // FFS do we need c31 is to be applied?

    //Create and configure cell

    f_EUTRA_CellConfig_Def(eutra_Cell1);

    // FFS - do we need to attach on EUTRA too?

    f_EUTRA_NR_WaitForCoOrd_Trigger (NR, "Preamble");

    // FFS Prose doesn't specify any power levels

    f_EUTRA38_SetCellPower(eutra_Cell1, tsc_ServingCellRS_EPRE, tsc_ServingCellRS_EPRE); //  FFS

    f_EUTRA_TestBody_Set(true);

    fl_TC_8_1_4_2_1_1_NR5GC_EUTRA_TestBody ();

    f_EUTRA_TestBody_Set(false);

    f_EUTRA_Postamble(eutra_Cell1, E2_CONNECTED, EPS_Only);

    // take down the NR cell

    f_EUTRA_NR_SendCoOrd (NR, cms_EUTRA_NR_Trigger("Postamble"));

  }
4.8 Change 8
	Function name
	Template cdr_TAU_Request_N1ToS1

	Reason for change
	For TRACKING AREA UPDATE REQUEST following 5GC->EPC handover, “type of security context” will be “mapped security context”. Also additionalGUTI and NonCurrentNativeNasKeySetId may be included


	Summary of change
	Change the checking of asmeKeySetId and allow other IEs omitted in the base template.

	TTCN module
	EUTRA38_NR5GCProcedures

	MCC160 Comment
	


Before change

      template (present) NAS_UL_Message_Type cdr_TAU_Request_N1ToS1(template (present)  EPS_UpdateTypeValue p_EpsUpdate_TypeValue,

                                                                    NAS_KsiValue p_KsiValue,

                                                                    template DRXparameter p_DRXparameter,

                                                                    template AdditionalUpdateType p_AdditionalUpdateType,

                                                                    template (present) MobileIdentity p_MobileId := cr_MobileIdAny,

                                                                    template TrackingAreaId p_TAI := cr_TAI_Any ifpresent)
modifies cr_508_TAU_Request :=

  {

    tRACKING_AREA_UPDATE_REQUEST := {

epsUpdateType              := cr_EPS_UpdateType(tsc_EpsUpdate_NotActive, p_EpsUpdate_TypeValue),

      oldGuti                    := p_MobileId,

      ueNetworkCapability        := cdr_NR5GC_UENetworkCap('58'O),

      lastVisitedRegisteredTai   := p_TAI,

      ueRadioCapNeeded           := cr_UeRadioCap,

      epsBearerContextStatus     := cr_EPS_BearerContextStatus_Any,

      oldGUTI_Type               := cr_GUTI_Type_Native,

      ueAddSecurityCap           := cr_UE_AddSecurityCap_Any,

      ueStatus                   := cr_UEStatus_N1

}

  };  
After change

  template (present) NAS_UL_Message_Type cdr_TAU_Request_N1ToS1(template (present)  EPS_UpdateTypeValue p_EpsUpdate_TypeValue,

                                                                    NAS_KsiValue p_KsiValue,

                                                                    template DRXparameter p_DRXparameter,

                                                                    template AdditionalUpdateType p_AdditionalUpdateType,

                                                                    template (present) MobileIdentity p_MobileId := cr_MobileIdAny,

                                                                    template TrackingAreaId p_TAI := cr_TAI_Any ifpresent)
    modifies cr_508_TAU_Request :=

  {

    tRACKING_AREA_UPDATE_REQUEST :=  {

      asmeNasKeySetId            := cr_NAS_KeySetIdentifier(p_KsiValue, tsc_NasKsi_MappedSecurityContext), 
      epsUpdateType              := cr_EPS_UpdateType(tsc_EpsUpdate_NotActive, p_EpsUpdate_TypeValue),

      oldGuti                    := p_MobileId,
      nonCurrentNativeNasKeySetId  := *,

      additionalGuti             :=  *,
      ueNetworkCapability        := cdr_NR5GC_UENetworkCap('58'O),

      lastVisitedRegisteredTai   := p_TAI,

      epsBearerContextStatus     := cr_EPS_BearerContextStatus_Any,

      oldGUTI_Type               := cr_GUTI_Type_Native,

      ueAddSecurityCap           := cr_UE_AddSecurityCap_Any,

      ueStatus                   := cr_UEStatus_N1

    }

  };
4.9 Change 9
	Function name
	f_EUTRA38_TrackingAreaUpdateFromN1_WithoutReq()

	Reason for change
	Following handover from 5GNR -> EUTRA, SRB2 will be active and so will be used for the Tracking Area Update signalling 

	Summary of change
	Use SRB2 when calling f_EUTRA_TrackingAreaUpdate_Accept_Complete  (see change 1)

	TTCN module
	EUTRA38_NR5GCProcedures

	MCC160 Comment
	


Before change

      function f_EUTRA38_TrackingAreaUpdateFromN1_WithoutReq(EUTRA_CellId_Type p_CellId,

                                                         RRC_TransactionIdentifier p_RRC_TI,

                                                         NAS_MSG_Indication_Type p_NasInd,

                                                         N1ToS1_SecurityContext_Type p_CxtType := MAPPED_CXT,

                                                         RRCConnectionReleaseRequired_Type p_ReleaseRequired := rrcConnectionRelease,

                                                         template (omit) PLMN_List p_EPLMNlist := omit) runs on EUTRA_PTC return NasCount_Type

  {

    var EUTRA_SecurityParams_Type v_SecurityParams := f_EUTRA_Security_Get();

    var GutiParameters_Type v_GutiParams:= f_EUTRA_CellInfo_GetGuti(p_CellId);

    var NAS_PlmnId v_PLMN := f_Asn2Nas_PlmnId (v_GutiParams.PLMN_Identity);

    if (p_CxtType == NEW_CXT) {

      //4a1 4a2: Authentication Procedure

      /* Authentication Procedure according to 24.301 cl. 5.4.2 */

      v_SecurityParams := f_EUTRA_NAS_Authentication(p_CellId,

                                                     tsc_SRB1,

                                                     v_SecurityParams,

                                                     tsc_SHT_NoSecurityProtection,

                                                     (tsc_SHT_NoSecurityProtection, tsc_SHT_IntegrityProtected),

                                                     v_PLMN);

      f_EUTRA_Security_Set(v_SecurityParams);

    }

    if (p_CxtType == NEW_CXT or p_CxtType == EXISTING_CXT) {

      // 4a3 4a4/4b1 4b2

      f_EUTRA_NAS_ActivateSecurity(p_CellId, v_SecurityParams);

    }

    /* Step 4 - 5: send TAU ACCEPT and receive TAU COMPLETE */

    f_EUTRA_TrackingAreaUpdate_Accept_Complete(p_CellId,

                                               v_GutiParams,

                                               tsc_EpsUpdate_TaUpdate,

                                               p_NasInd.Pdu.Msg.tRACKING_AREA_UPDATE_REQUEST.epsBearerContextStatus,

                                               p_NasInd.Pdu.Msg.tRACKING_AREA_UPDATE_REQUEST.addUpdateType,

                                               NORMAL,

                                               p_EPLMNlist,

                                               cs_EPS_NwkFtSup_NR);

      /* Step 8: Release RRC connection */

      if (p_ReleaseRequired == rrcConnectionRelease) {

        f_EUTRA_RRC_ConnectionRelease(p_CellId, p_RRC_TI);

      }

    return p_NasInd.SecurityProtection.NasCount;

  }

After change

  function f_EUTRA38_TrackingAreaUpdateFromN1_WithoutReq(EUTRA_CellId_Type p_CellId,

                                                         RRC_TransactionIdentifier p_RRC_TI,

                                                         NAS_MSG_Indication_Type p_NasInd,

                                                         N1ToS1_SecurityContext_Type p_CxtType := MAPPED_CXT,

                                                         RRCConnectionReleaseRequired_Type p_ReleaseRequired := rrcConnectionRelease,

                                                         template (omit) PLMN_List p_EPLMNlist := omit) runs on EUTRA_PTC return NasCount_Type

  {

    var EUTRA_SecurityParams_Type v_SecurityParams := f_EUTRA_Security_Get();

    var GutiParameters_Type v_GutiParams:= f_EUTRA_CellInfo_GetGuti(p_CellId);

    var NAS_PlmnId v_PLMN := f_Asn2Nas_PlmnId (v_GutiParams.PLMN_Identity);

    if (p_CxtType == NEW_CXT) {

      //4a1 4a2: Authentication Procedure

      /* Authentication Procedure according to 24.301 cl. 5.4.2 */

      v_SecurityParams := f_EUTRA_NAS_Authentication(p_CellId,

                                                     tsc_SRB1,

                                                     v_SecurityParams,

                                                     tsc_SHT_NoSecurityProtection,

                                                     (tsc_SHT_NoSecurityProtection, tsc_SHT_IntegrityProtected),

                                                     v_PLMN);

      f_EUTRA_Security_Set(v_SecurityParams);

    }

    if (p_CxtType == NEW_CXT or p_CxtType == EXISTING_CXT) {

      // 4a3 4a4/4b1 4b2

      f_EUTRA_NAS_ActivateSecurity(p_CellId, v_SecurityParams);

    }

    /* Step 4 - 5: send TAU ACCEPT and receive TAU COMPLETE */

    f_EUTRA_TrackingAreaUpdate_Accept_Complete(p_CellId,

                                               v_GutiParams,

                                               tsc_EpsUpdate_TaUpdate,

                                               p_NasInd.Pdu.Msg.tRACKING_AREA_UPDATE_REQUEST.epsBearerContextStatus,

                                               p_NasInd.Pdu.Msg.tRACKING_AREA_UPDATE_REQUEST.addUpdateType,

                                               NORMAL,

                                               p_EPLMNlist,

                                               cs_EPS_NwkFtSup_NR, true);

      /* Step 8: Release RRC connection */

      if (p_ReleaseRequired == rrcConnectionRelease) {

        f_EUTRA_RRC_ConnectionRelease(p_CellId, p_RRC_TI);

      }

    return p_NasInd.SecurityProtection.NasCount;

  }

4.10 Change 10
	Function name
	f_EUTRA38_MappedContextFromN1_InitNAS()

	Reason for change
	KeNB needs to be refreshed before deriving RRC encryption and integrity keys according 33.501

	Summary of change
	Refresh KeNB  using new function (see change 12)

	TTCN module
	EUTRA38_NR5GCProcedures

	MCC160 Comment
	


Before change

 function f_EUTRA38_MappedContextFromN1_InitNAS(EUTRA_CellId_Type p_CellId,

                                               EUTRA_NR_CoOrd_Security_Type p_NR_Auth_Params,

                                               boolean p_Handover )  runs on EUTRA_PTC return EUTRA_SecurityParams_Type

  {

    var EUTRA_SecurityParams_Type v_SecurityParams := f_EUTRA_Security_Get();

    var NasCount_Type v_NasCountUL;

    v_SecurityParams.KASME := f_NG_Authentication_A14(p_Handover,

                                                      p_NR_Auth_Params.NASCount,

                                                      p_NR_Auth_Params.Key,

                                                      v_SecurityParams.KDF);

    // Derives KNASenc key

    v_SecurityParams.NAS_Ciphering.K_NAS := f_EUTRA_NB_Authentication_S15 (tsc_NAS_Enc_Alg,v_SecurityParams.NAS_Ciphering.Algorithm,

                                                                           v_SecurityParams.KASME,

                                                                           v_SecurityParams.KDF);

    // Derives KNASint key

    v_SecurityParams.NAS_Integrity.K_NAS := f_EUTRA_NB_Authentication_S15 (tsc_NAS_Int_Alg,v_SecurityParams.NAS_Integrity.Algorithm,

                                                                           v_SecurityParams.KASME,

                                                                           v_SecurityParams.KDF);

    v_SecurityParams.KSIsgsn := v_SecurityParams.AuthParams.KeySeq ;

    // Derives KNASint key

    //As per 33.401 clause 9.2.2.2

    v_NasCountUL := 'FFFFFFFF'O;

    v_SecurityParams.KENB := f_EUTRA_NB_Authentication_S11 (v_SecurityParams.KDF, v_SecurityParams.KASME, v_NasCountUL); // @sic R5s160711 sic@

    //Derive KenB

    v_SecurityParams.AS_Ciphering.KRRCenc := f_EUTRA_NB_Authentication_S15 (tsc_RRC_Enc_Alg,

                                                                       f_EUTRA_NB_ConvertCiphAlg2Bitstring(v_SecurityParams.AS_Ciphering.Algorithm),

                                                                       v_SecurityParams.KENB,

                                                                       v_SecurityParams.KDF);

    // Derives KRRCenc key

    v_SecurityParams.AS_Integrity.KRRCint := f_EUTRA_NB_Authentication_S15 (tsc_RRC_Int_Alg,

                                                                       fl_AS_IntegrityInfo_GetIntAlg(v_SecurityParams.AS_Integrity),  // @sic R5-126020 sic@

                                                                       v_SecurityParams.KENB,

                                                                       v_SecurityParams.KDF);

    // Derives KRRCenc key

    v_SecurityParams.AS_Ciphering.KUPenc := f_EUTRA_NB_Authentication_S15 (tsc_UP_Enc_Alg,

                                                                      f_EUTRA_NB_ConvertCiphAlg2Bitstring(v_SecurityParams.AS_Ciphering.Algorithm),

                                                                      v_SecurityParams.KENB,

                                                                      v_SecurityParams.KDF);
    // Now update the SS

    f_EUTRA_NB_SS_NAS_ConfigureSecurity(NASCTRL, v_SecurityParams.NAS_Integrity, v_SecurityParams.NAS_Ciphering); // @sic R5s120686 sic@

    f_EUTRA_NB_SS_NAS_CountSet(NASCTRL, '00000000'O, omit);

    f_EUTRA_SS_RRC_EnableIntProt_CiphULandDL(p_CellId,

                                             v_SecurityParams.AS_Integrity,

                                             v_SecurityParams.AS_Ciphering,

                                             cs_TimingInfo_Now);

    return v_SecurityParams;

  }  
After change

  function f_EUTRA38_MappedContextFromN1_InitNAS(EUTRA_CellId_Type p_CellId,

                                               EUTRA_NR_CoOrd_Security_Type p_NR_Auth_Params,

                                               boolean p_Handover )  runs on EUTRA_PTC return EUTRA_SecurityParams_Type

  {

    var EUTRA_SecurityParams_Type v_SecurityParams := f_EUTRA_Security_Get();

    var NasCount_Type v_NasCountUL;

    //KS R5sxxxx

    var NextHopChainingCount v_NCC :=  2;

    var CellCarrierFreqEUTRA_Type v_EARFCN := f_EUTRA_CellInfo_GetEARFCN(p_CellId);

    var PhysCellId  v_TargetPhysicalCellIdentity := f_EUTRA_CellInfo_GetPhyCellId(p_CellId);

    v_SecurityParams.KASME := f_NG_Authentication_A14(p_Handover,

                                                      p_NR_Auth_Params.NASCount,

                                                      p_NR_Auth_Params.Key,

                                                      v_SecurityParams.KDF);

    // Derives KNASenc key

    v_SecurityParams.NAS_Ciphering.K_NAS := f_EUTRA_NB_Authentication_S15 (tsc_NAS_Enc_Alg,v_SecurityParams.NAS_Ciphering.Algorithm,

                                                                           v_SecurityParams.KASME,

                                                                           v_SecurityParams.KDF);

    // Derives KNASint key

    v_SecurityParams.NAS_Integrity.K_NAS := f_EUTRA_NB_Authentication_S15 (tsc_NAS_Int_Alg,v_SecurityParams.NAS_Integrity.Algorithm,

                                                                           v_SecurityParams.KASME,

                                                                           v_SecurityParams.KDF);

    v_SecurityParams.KSIsgsn := v_SecurityParams.AuthParams.KeySeq ;

    // Derives KNASint key

    //As per 33.401 clause 9.2.2.2

    v_NasCountUL := 'FFFFFFFF'O;

    v_SecurityParams.KENB := f_EUTRA_NB_Authentication_S11 (v_SecurityParams.KDF, v_SecurityParams.KASME, v_NasCountUL); // @sic R5s160711 sic@

    //Derive KenB

    //KS R5sxxxxx

    v_SecurityParams := f_EUTRA_InitAS_KeyChaining_KeyRefresh_Handover_5GC2EPC(v_TargetPhysicalCellIdentity, 

                                                                               v_EARFCN.dl_CarrierFreq, 

                                                                               v_SecurityParams, 

                                                                               v_NCC); /* Calculate Kenb and KRRCenc, KRRCint, KUPenc */

    /* MOVED into new function v_SecurityParams.AS_Ciphering.KRRCenc := f_EUTRA_NB_Authentication_S15 (tsc_RRC_Enc_Alg,

                                                                       f_EUTRA_NB_ConvertCiphAlg2Bitstring(v_SecurityParams.AS_Ciphering.Algorithm),

                                                                       v_SecurityParams.KENB,

                                                                       v_SecurityParams.KDF);

    // Derives KRRCenc key

    v_SecurityParams.AS_Integrity.KRRCint := f_EUTRA_NB_Authentication_S15 (tsc_RRC_Int_Alg,

                                                                       fl_AS_IntegrityInfo_GetIntAlg(v_SecurityParams.AS_Integrity),  // @sic R5-126020 sic@

                                                                       v_SecurityParams.KENB,

                                                                       v_SecurityParams.KDF);

    // Derives KRRCenc key

    v_SecurityParams.AS_Ciphering.KUPenc := f_EUTRA_NB_Authentication_S15 (tsc_UP_Enc_Alg,

                                                                      f_EUTRA_NB_ConvertCiphAlg2Bitstring(v_SecurityParams.AS_Ciphering.Algorithm),

                                                                      v_SecurityParams.KENB,

                                                                      v_SecurityParams.KDF);

                                                                      */

    // Now update the SS

    f_EUTRA_NB_SS_NAS_ConfigureSecurity(NASCTRL, v_SecurityParams.NAS_Integrity, v_SecurityParams.NAS_Ciphering); // @sic R5s120686 sic@

    f_EUTRA_NB_SS_NAS_CountSet(NASCTRL, '00000000'O, omit);

    f_EUTRA_SS_RRC_EnableIntProt_CiphULandDL(p_CellId,

                                             v_SecurityParams.AS_Integrity,

                                             v_SecurityParams.AS_Ciphering,

                                             cs_TimingInfo_Now);

    return v_SecurityParams;

  }
4.11 Change 11
	Function name
	New function f_EUTRA_InitAS_KeyChaining_KeyRefresh_Handover_5GC2EPC()

	Reason for change
	KeNB needs to be refreshed before deriving RRC encryption and integrity keys according 33.501

	Summary of change
	New function based on function f_EUTRA_InitAS_KeyChaining_KeyRefresh()

	TTCN module
	

	MCC160 Comment
	


New function

function f_EUTRA_InitAS_KeyChaining_KeyRefresh_Handover_5GC2EPC(PhysCellId p_TargetPhysicalCellIdentity,

                                                 ARFCN_ValueEUTRA p_EARFCN_DL,  // @sic R5s100138 sic@

                                                 EUTRA_SecurityParams_Type p_Auth_Params,

                                                 NextHopChainingCount p_NCC) return EUTRA_SecurityParams_Type

  { // @sic R5s100138 sic@

    var EUTRA_SecurityParams_Type v_Auth_Params := p_Auth_Params;

    while (v_Auth_Params.NCC != p_NCC)  {



   if (v_Auth_Params.NCC == 0) { 



     v_Auth_Params.NH := f_EUTRA_NB_Authentication_S12(v_Auth_Params.KDF,



                                                            v_Auth_Params.KASME,



                                                            v_Auth_Params.KENB); 



    }



    else  { // NH for NCC =1 is already derived in initial AS Key derivation

          v_Auth_Params.NH := f_EUTRA_NB_Authentication_S12(v_Auth_Params.KDF,

                                                            v_Auth_Params.KASME,

                                                            v_Auth_Params.NH); 

        }

        v_Auth_Params.NCC := (v_Auth_Params.NCC + 1) mod 4;

    }  

    v_Auth_Params.KENB := f_EUTRA_NB_Authentication_S13(v_Auth_Params.KDF,

                                                        v_Auth_Params.KENB,

                                                        v_Auth_Params.NH,

                                                        p_TargetPhysicalCellIdentity,

                                                        p_EARFCN_DL, true);

    //Derive KENB* which will be KENB from NH stored in v_Auth_Params.KENB

    v_Auth_Params.AS_Ciphering.KRRCenc := f_EUTRA_NB_Authentication_S15(tsc_RRC_Enc_Alg,

                                                                        f_EUTRA_NB_ConvertCiphAlg2Bitstring(v_Auth_Params.AS_Ciphering.Algorithm),

                                                                        v_Auth_Params.KENB,

                                                                        v_Auth_Params.KDF);

    // Derives KRRCenc key

    v_Auth_Params.AS_Integrity.KRRCint := f_EUTRA_NB_Authentication_S15(tsc_RRC_Int_Alg,

                                                                       fl_AS_IntegrityInfo_GetIntAlg(v_Auth_Params.AS_Integrity),  // @sic R5-126020 sic@

                                                                       v_Auth_Params.KENB,

                                                                       v_Auth_Params.KDF);

    // Derives KRRCenc key

    v_Auth_Params.AS_Ciphering.KUPenc := f_EUTRA_NB_Authentication_S15 (tsc_UP_Enc_Alg,

                                                                      f_EUTRA_NB_ConvertCiphAlg2Bitstring(v_Auth_Params.AS_Ciphering.Algorithm),

                                                                      v_Auth_Params.KENB,

                                                                      v_Auth_Params.KDF);

    // Derives KRRCenc key;

    // Ciphering algorithm for UP is same as RRC

    return v_Auth_Params;

  };

4.12 Change 12
	Function name
	Template cr_UEStatus_S1

	Reason for change
	IEI of UEStatus in REGISTRATION REQUEST is 0x2B

	Summary of change
	Correct IEI

	TTCN module
	NR5GC_IRAT_CommonFunctions

	MCC160 Comment
	


Before change

  template (present) UE_Status cr_UEStatus_S1 :=

  { // 24.501 cl. 9.11.3.56

    iei     := '6D'O,

    iel     := ?,

    spare   := tsc_Spare6,

    n1ModeRegistered    := ?,

    s1ModeRegistered    := '1'B

  };

After change

  template (present) UE_Status cr_UEStatus_S1 :=

  { // 24.501 cl. 9.11.3.56
    iei     := '2B'O,
    iel     := ?,

    spare   := tsc_Spare6,

    n1ModeRegistered    := ?,

    s1ModeRegistered    := '1'B

  };
5. Execution Log Files

5.1 HiSilicon Balong 5000
The HiSilicon Balong 5000UE passed this test case on Keysight 5G Protocol Conformance Toolset Solution on NR5GC band n41.The documentation below is enclosed as evidence of the successful test case run [1]:

· Test Case Execution log file: 
TC_8_1_4_2_1_1_LOG.html
In the log file (in .html format) the complete test case execution can be seen. All TLI events are presented and message contents are fully decoded and can be verified. Preliminary verdicts and the final test case verdict can be seen in the log file.

